

## L'objet Page

Avant d'énumérer des membres de composants, il est intéressant d'en examiner quelques uns de la page elle-même. L'objet **Page** est l'équivalent en Web du **Form** de Windows. Comme cela a déjà été illustré, il est décrit en HTML dans le fichier .aspx dont l'accès à la partie visible par l'onglet *Design* permet d'y glisser les composants souhaités. La programmation sous jacente à cette page est écrite dans le fichier dont l'extension est .aspx.vb. L'interface de programmation est identique à celle utilisée pour la programmation événementielle des Windows Forms. Une liste déroulante présente les objets et une autre présente les événements pouvant être émis par l'objet sélectionné.

Parmi les événements auxquels peut répondre le code de l'objet **Page**, ceux liés aux échanges entre le navigateur et le serveur sont les plus importants. Ils sont au nombre de dix et offrent autant d'occasions d'interventions au programmeur. Ils sont énumérés ici dans leur ordre d'émission.

Les événements **PreInit**, **Init**, **InitComplete**, **PreLoad**, **Load** et **LoadComplete** sont émis pendant le chargement de l'objet **Page** par le serveur (l'objet **Page** contient le code que doit exécuter le serveur). C'est entre les événements **Load** et **LoadComplete** que sont exécutées les éventuelles procédures événementielles liées aux composants visuels.

Les événements **PreRender**, **PreRenderComplete**, **SaveStateComplete** et **Unload** sont émis lors de l'expédition de la page au navigateur, **PreRender** étant l'initiation de l'envoi et **Unload** constituant l'abandon de l'objet **Page** par le serveur qui termine ainsi sa session de travail.

Les méthodes de l'objet **Page** sont nombreuses et nul doute que le *programmeur Web* y trouvera son bonheur. Pour ce qui concerne ce cours, seule la méthode **SetFocus** est retenue. Elle permet de désigner le composant qui doit avoir la main à l'issue de l'affichage de la page par le navigateur. Elle peut être programmée dans n'importe quelle partie du code pour autant qu'il soit exécuté avant l'événement **PreRenderComplete**.

```
Page.SetFocus(Bouton)           ' Donne le focus au composant nommé Bouton
```

Les trois propriétés **Form**, **IsPostBack** et **EnableViewState** sont à épinglez ici.

**Form** Cette propriété de l'objet **Page** contient le formulaire HTML de la page et donc, tous les objets présents sur le Web Form.

```
For Each X As Control In Page.Form.Controls
    UneListe.Items.Add(X.ID)           ' Remplit une ListBox nommée UneListe avec les noms
Next                                   ' des composants ASP présents sur la Web Form
```

**IsPostBack** Cette propriété en lecture seule reçoit la valeur **False** lors du premier affichage la page et la valeur **True** quand la page est une réponse en retour au serveur. La consultation de cette propriété dans une alternative permet de séparer les tâches à accomplir lors d'un premier affichage de celles à accomplir à chaque affichage.

```
Private Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs)
    Handles Me.Load
    If Me.IsPostBack Then
        ' Tâches à réaliser pour répondre aux demandes de l'internaute.
    Else
        ' Initialisations diverses et tâches à réaliser avant le premier envoi au navigateur.
    End If
End Sub
```

**EnableViewState** Cette propriété autorise la persistance des valeurs des propriétés de l'objet (l'objet **Page** dans ce cas) pour lequel elle est reçoit la valeur **True** et l'interdit dans le cas contraire. Sa valeur par défaut est toujours **True**. Si la valeur **False** est affectée à cette propriété de l'objet **Page**, la persistance des propriétés des composants n'est plus possible. Les propriétés ainsi conservées doivent avoir été *déclarées* dans un objet **StateBag** qui est automatiquement mis en place par la commande **ViewState**. Cette commande stocke sous la référence d'un indice ou d'un libellé la propriété qui lui est désignée d'un l'objet. Au niveau de chaque composant, elle peut être testée et modifiée selon les besoins du programmeur. Là aussi, sa valeur par défaut est **True** et cette propriété d'un composant doit être traitée comme toute autre propriété dont la persistance est nécessaire car tant que les

propriétés conservées ne sont pas restaurées, elles reçoivent toutes leur valeur par défaut au chargement d'un objet **Page**. L'exemple suivant montre la récupération d'une propriété d'un composant et celle d'une variable précédemment sauvegardées.

```
UnBooleen = ViewState("UnBooleen")
TXT.Text = ViewState("TXTText")
```

Ci-dessus, la référence passée à la commande **ViewState** représente le nom de la variable stockée ou celui d'une propriété concaténée au nom du composant auquel elle appartient. Cette pratique améliore la lisibilité du code.

```
Partial Class _Default
```

```
    Inherits System.Web.UI.Page
    Dim UnEntier As Integer = 0
    Dim UnBooleen As Boolean
```

```
    Private Sub Page_Load(ByVal sender As Object, ByVal e As System.EventArgs)
                                                Handles Me.Load
        Page.EnableViewState = True           ' C'est la valeur par défaut
        If Me.IsPostBack Then                 ' La page est revenue et il faut répondre
            UnEntier = ViewState("UnEntier")  ' Récupération des valeurs
            UnBooleen = ViewState("UnBooleen")
            Etiquette.EnableViewState = ViewState("EtiquetteEnableViewState")
            TXT.EnableViewState = ViewState("TXTEnableViewState")
            If Etiquette.EnableViewState Then
                Etiquette.BackColor = ViewState("EtiquetteBackColor")
                Etiquette.ForeColor = ViewState("EtiquetteForeColor")
            End If
            If TXT.EnableViewState Then
                TXT.Text = ViewState("TXTText")
                TXT.BackColor = ViewState("TXTBackColor")
                TXT.ForeColor = ViewState("TXTForeColor")
            End If
            UnEntier += 1
            UnBooleen = Not UnBooleen
            TXT.Text = UnEntier
            Etiquette.Text = UnBooleen.ToString
        Else                                  ' La page est envoyée pour la première fois
            Etiquette.Text = "MonTexte"
            TXT.Text = UnEntier
            UnBooleen = True
        End If
    End Sub
```

```
    Protected Sub Bouton_Click(ByVal sender As Object, ByVal e As System.EventArgs)
                                                Handles Bouton.Click
        TXT.EnableViewState = True           ' C'est la valeur par défaut
        Etiquette.EnableViewState = Not Etiquette.EnableViewState
        Select Case UnEntier
            Case 1
                Etiquette.BackColor = Drawing.Color.Cyan
                Etiquette.ForeColor = Drawing.Color.Brown
            Case 2
                TXT.BackColor = Drawing.Color.Green
                TXT.ForeColor = Drawing.Color.Yellow
            Case 3
                Etiquette.BackColor = Drawing.Color.Black
                Etiquette.ForeColor = Drawing.Color.White
                TXT.BackColor = Drawing.Color.White
                TXT.ForeColor = Drawing.Color.Magenta
            Case Else
                UnEntier = 0
        End Select
    End Sub
    Protected Sub BImage_Click(ByVal sender As Object,
                                ByVal e As System.Web.UI.ImageClickEventArgs) Handles BImage.Click
        ' Ne rien faire ici pour l'instant. Le clic de BImage montre l'évolution de UnEntier dans
```

```
' TXT et la persistance des propriétés conservées.
End Sub

Protected Sub Page_PreRender(ByVal sender As Object, ByVal e As System.EventArgs)
    Handles Me.PreRender
    ViewState("UnEntier") = UnEntier          ' Sauvegarde des valeurs
    ViewState("UnBooleen") = UnBooleen
    ViewState("EtiquetteEnableViewState") = Etiquette.EnableViewState
    ViewState("EtiquetteText") = Etiquette.Text
    ViewState("EtiquetteBackColor") = Etiquette.BackColor
    ViewState("EtiquetteForeColor") = Etiquette.ForeColor
    ViewState("TXTEnableViewState") = TXT.EnableViewState
    ViewState("TXTText") = TXT.Text
    ViewState("TXTBackColor") = TXT.BackColor
    ViewState("TXTForeColor") = TXT.ForeColor
End Sub
End Class
```